/*
Copyright (c) 2009, Young Digital Planet SA
All rights reserved.

* Redistribution and use in source and binary forms, with or without 
  modification, are permitted provided that the following conditions are met:
* Redistributions of source code must retain the above copyright notice, this 
  list of conditions and the following disclaimer.
* Redistributions in binary form must reproduce the above copyright notice, 
  this list of conditions and the following disclaimer in the documentation 
  and/or other materials provided with the distribution.
* Neither the name of the Young Digital Planet SA nor the names of its 
  contributors may be used to endorse or promote products derived from this 
  software without specific prior written permission.

THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" 
AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE 
IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE 
ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE 
LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR 
CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF 
SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS 
INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN 
CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE) 
ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE 
POSSIBILITY OF SUCH DAMAGE.
*/
package pl.ydp.utils
{
	/**
	 * Umożliwia czytanie stringu znak po znaku<br>
	 * Użycie:
	 * <pre>
	 * a = new StringCharsReader( "abcdef" );
	 * while( a.hasNext() )
	 * 	trace( a.nextChar() )
	 * </pre>
	 */
	public class StringCharsReader
	{
		/**
		 * Czytany string
		 */
		protected var m_str:String;
		/**
		 * Indekx następnego znaku
		 */
		protected var m_next:int = 0;
		
		/**
		 * Utworzenie readera ze wskazanego string'u
		 */
		public function StringCharsReader( str:String ) {
			m_str = str;
		}
		
		/**
		 * Sprawdza czy są jeszcze znaki do odczytania
		 * @return true gdy są jeszcze znaki do odczytania
		 */ 
		public function get hasNext():Boolean {
			return m_next<m_str.length;
		}
		/**
		 * Pobiera następny znak string'a, przesuwająć wskaźnik czytania
		 * na następny znak.
		 * @return następny znak
		 * @throws pl.ydp.utils.NoSuchElementError gdy nie ma następnego znaku do 
		 * odczytania
		 */
		public function nextChar():String {
			if ( m_next>=m_str.length )
				throw new NoSuchElementError( "nextChar" );
			return m_str.charAt( m_next++ );
		}
		
		/**
		 * Pobiera następny znak string'a bez przesuwania
		 * wskaźnika następnego znaku, następne wołania nextChar, peekNextChar
		 * zwrócą ten sam znak.
		 * @return następny znak
		 * @throws pl.ydp.utils.NoSuchElementError gdy nie ma następnego znaku do 
		 * odczytania
		 */
		public function peekNextChar():String {
			if ( m_next>=m_str.length )
				throw new NoSuchElementError( "nextChar" );
			return m_str.charAt( m_next );
		}
		
		/**
		 * Ominięcie 'białych' znaków - spacji, tabulatora, \n, \r
		 */
		public function skipWhite():void {
			while( m_next<m_str.length ) {
				var nch:String =  m_str.charAt( m_next );
				if ( nch==" " || nch=="\t" || nch=="\n" || nch=="\r" )
					++m_next;
				else
					break;
			}
		}
		
		/**
		 * Returns begin index and end index of a word which 
		 * given index is inside.
		 */
		public function getWordBounds(ind:int):Array{
			var begInd:int = ind;
			var endInd:int = ind;
			
			while(m_str.charAt(begInd) != " " && begInd > 0)
				begInd--;
				
			while(m_str.charAt(endInd) != " " && endInd < m_str.length)
				endInd++;
			
			if(m_str.charAt(begInd) == " ")
				begInd++;
			
			return [begInd, endInd];
		}
		
		/**
		 * Odczytuje string abc lub "a\"bc". Gdy string jest w cudzysłowach, to 
		 * one wyznaczają początek i koniec. Terminator nie jest wtedy
		 * używany
		 * @param terminator informacja o tym kiedy kończy się string
		 * gdy jest to String to znak oznacza to znak terminujący, albo
		 * <pre>function isTerminator( char:String ):Boolean;</pre>, albo object
		 * dla w którym jest sprawdzanym czy dla danego znaku jest przypisany
		 * element!=null (wartości zero i false też oznaczają terminator)
		 * Terminator nie jest odczytywany z reader'a.
		 * @return odczytany string
		 */
		public function readString( terminator:* ):String {
			if ( !hasNext )
				return "";
			if ( peekNextChar()=="\"" )
				return readQuotedString();
			else {
				var res:Array = [];
				
				loadChars:
				do {
					var char:String = peekNextChar();
					if ( ((terminator is String) ? 
							char==terminator : 
							(terminator is Function ?
								terminator(char) :
								terminator[char]!=null)) )
						break loadChars;
					res.push( nextChar() );
				} while( hasNext );
				
				return res.join( "" );
			}
		}
		
		/**
		 * Odczytuje string zamknięty w cudzysłowia (f-ja ta zakłada
		 * że pierwszy znak jest na pewno cudzysłowem i go usuwa)
		 * @return odczytany string
		 */
		protected function readQuotedString():String {
			nextChar();	// usuwam cudzysłów
			
			var res:Array = [];
				
			try {
				loadChars:
				do {
					var char:String = nextChar();
					if ( char=="\"" )
						break loadChars;
					if( char=="\\" ) {
						char = nextChar();
						if ( char=="n" )
							char = "\n";
						if ( char=="r" )
							char = "\r";
						if ( char=="t" )
							char = "\t";
					}
					res.push( char );
				} while( true );
			} catch( x:NoSuchElementError ) {
				throw new NoSuchElementError( "Unterminated string \""+res.join("")+"\"" );
			}
			
			return res.join( "" );
		}
	}
}